IDENTIFICATION

SPLINE CURVE INTERPOLATION - MU KINT
704 FORTRAN SUBROUTINE
Richard S. Christian, Terry W. Edwards - Revised October 20, 1964
Midwestern Universities Research Association, Stoughton, Wisconsin

PURPOSE

To interpolate coordinate values, derivatives, and arc lengths along a plane curve y(x) for specified x_o or y_o; the curve being one of M coordinate points $(x,y)_i$ previously processed by MU KUFIT.

METHOD

From tables of coordinates x_i and y_i and from the associated table of coefficients C_{1i}, C_{2i}, and C_{3i}, and arc lengths C_{4i}, valid in the interval $(x,y)_i$ to $(x,y)_{1+i}$, such that

$$\Delta y = C_{1i}\,\Delta x + C_{2i}\Delta x^2 + C_{3i}\Delta x^3 \quad \text{if } \Delta y_{i+1} \leq \Delta x_{i+1},$$

or $$\Delta x = C_{1i}\,\Delta y + C_{2i}\Delta y^2 + C_{3i}\Delta y^3 \quad \text{if } \Delta y_{i+1} > \Delta x_{i+1},$$

where $\Delta x = (x - x_i)$ and $\Delta y = (y - y_i)$,

MU KINT can obtain, for specified x_o or y_o, interpolated values of $y(x_o)$, $y'(x_o)$, $y''(x_o)$, or of $x(y_o)$, $x'(y_o)$, $x''(y_o)$, respectively, together with associated arc lengths.

If it is required, MU KINT will invert (analytically) one of the above cubic equations, in order to achieve a solution. Therefore, either x_o or y_o, may be specified upon entry into MU KINT.

USAGE

Any one entry into MU KINT may result in multiple sets of interpolated coordinates, derivatives, and arc lengths. Such multiple sets will be ordered in the output array (see below) in terms of increasing arc length, even if more than one such set is found to lie between $(x,y)_i$ and $(x,y)_{i+1}$.

Limits may be set on the portion of coordinate and coefficient arrays, to be searched over, and/or the number of sets of interpolated answers may be limited so as to not exceed the users storage allocation.

MU KINT may be entered via the 14 argument calling sequence:

CALL KINT (X,Y,CO,CRV,M,XO,YO,IXY,KMIN,KMAX,KANS, KFMIN,KFMAX,KFANS

where the various arguments have the following significance.

X: One dimensional array of x coordinates specifying curve.
Y: One dimensional array of y coordinates specifying curve.

CO: A two dimensional array, CO(4,...) of coefficients and arc lengths generated by MU KUFIT.

CRV: A two dimensional array, CRV(6...), of output (interpolated) coordinates, derivatives, and arc lengths:
CRV(1,n): x coordinate of n^{th} interpolation.
CRV(2,n): y coordinate of n^{th} interpolation.
CRV(3,n): Signal for which derivatives are returned - 1: y'(x) and y''(x), 2: x'(y) and x''(y).
CRV(4,n): First derivative at n^{th} interpolation.
CRV(5,n): Second derivative at n^{th} interpolation.
CRV(6,n): Arc length derivative at n^{th} interpolation.

M: Number of x and y coordinate pairs comprising curve. This specifies the length of the curve, and must not exceed the dimensions of X, Y, and CO.

XO: Specified x_o coordinate. Dummy if y_o is specified.
YO: Specified y_o coordinate. Dummy if x_o is specified.

IXY: Signal for coordinate specified: 1: XO specified, 2: YO specified. IXY is returned through CRV(3,n).

KMIN: Specified lower end of X, Y arrays to be searched.
KMAX: Specified upper end of X, Y arrays to be searched.

$$(1 \leqq \text{KMIN} < \text{KMAX} \leqq \text{M}),$$

KANS: Specified limiting number of interpolated sets answers for a single entry into MU KINT.*

*Note: It is possible to limit the number of sets of answers accepted (from a multivalued curve) from one entry into MU KINT. The answers returned through the CRV array will be no greater in number than KANS, and will be ordered in terms of arc length beginning at KMIN. The remaining portions of the curve (between KFMIN and KFMAX, see below) as well as between 1 and KMIN and between KMAX and M, will not have been searched upon exit from MU KINT.

KFMIN: Returned as lower end of X, Y arrays not searched.
KFMAX: Returned as upper end of X, Y arrays not searched.

$$(KMIN < KFMIN \leq KFMAX \leq KMAX).$$

KFANS: Returned as number of sets of solutions inserted into the CRV array.**

**Note: Another entry into MU KINT, with modified KMIN and KMAX, is required to further extend the searched portions of the arrays. When KFMIN is returned equal to KFMAX and KMAX, the searching of the arrays from KMIN to KMAX has been completed.

AUXILIARY ROUTINES

MU KINT requires the following subroutines:

1. Floating point SQRT (floating point argument).
2. MU CUBIC, the explicit solution of the general cubic equation.

CODING INFORMATION

MU KINT is available in both FORTRAN source card and binary card forms. Approximately 2300 locations of core storage is required.

```
C       SUBROUTINE KINT  -  CHRISTIAN,EDWARDS
C
C       TWO DIMENSIONAL SPLINE CURVE FITTING INTERPOLATION
C
C       PROGRAM TO INTERPOLATE COORDINATE VALUES AND ARC
C       LENGTHS ALONG TWO DIMENSIONAL CURVE, Y=Y(X),OF
C       EITHER CLOSED OR OPEN FORM. THE FITTED CURVE
C       MUST BE STORED IN A MANNER IDENTICAL WITH THAT
C       GENERATED BY SUBROUTINE KUFIT WRITTEN BY THE
C       SAME AUTHORS.
C
C       TWO DIFFERENT CALLING SEQUENCES ARE POSSIBLE.
C       METHOD 1 REQUIRES LESS PROGRAM STORAGE, METHOD 2
C       REQUIRES LESS WORK.
C
C       CALLING SEQUENCE 1 - USER MAY ALTER COMMON STORAGE.
C
C       SUBROUTINE KINT (MM,XO,YO,IIXY,KKMIN,KKMAX,KKANS,KFMIN,KFMAX,KFANS
C      1)
C       DIMENSION X(100),Y(100),CO(4,100),CRV(6,100)
C       COMMON X,Y,CO,CRV
C
C       CALLING SEQUENCE 2 - THIS IS NORMAL SEQUENCE.
C
        SUBROUTINE KINT (X,Y,CO,CRV,MM,XO,YO,IIXY,KKMIN,KKMAX,KKANS,KFMIN,
       1KFMAX,KFANS)
        DIMENSION X(1),Y(1),CO(4,1),CRV(6,1)
C
        DIMENSION AI(4),XI(3)
        M=MM
        INBR=20
        XNBR=INBR
        XX=XO
        YY=YO
        IXY=IIXY
        ZX=0.0
        ZY=0.0
        KMIN=KKMIN
        KMAX=KKMAX
        KANS=KKANS
        IF (KMIN) 10,10,12
 10     KMIN=1
 12     IF (KMAX-M) 16,16,14
 14     KMAX=M
 16     L=0
        DO 18 I=1,6
        DO 18 J=1,KANS
 18     CRV(I,J)=0.0
        K=KMIN
        IF (K-KMAX) 20,19,540
 19     K=K-1
 20     DXN=X(K+1)-X(K)
        DYN=Y(K+1)-Y(K)
        MTYPE=0
        DO 22 I=1,3
 22     XI(I)=0.0
        J=0
        DISC=0.0
        IF (IXY-1) 25,25,225
 25     ZX=XX
```

```
      XMIN=X(K)
      IF (X(K+1)-XMIN) 30,35,35
30    XMIN=X(K+1)
35    XMAX=X(K)
      IF (X(K+1)-XMAX) 45,40,40
40    XMAX=X(K+1)
45    IF (ABSF(DYN)-ABSF(DXN)) 130,130,50
50    IF (ABSF(CO(2,K))+ABSF(CO(3,K))) 130,130,52
52    IF (CO(3,K)) 54,58,54
54    DISC = CO(2,K)**2-3.0*CO(1,K)*CO(3,K)
      IF (DISC) 130,56,56
56    DY1 = (-CO(2,K)+SQRTF(DISC))/(3.0*CO(3,K))
      DY2 = (-CO(2,K)-SQRTF(DISC))/(3.0*CO(3,K))
      GO TO 60
58    DY1=-0.5*CO(1,K)/CO(2,K)
      DY2 = 0.0
60    Y1 = Y(K)+DY1
      Y2 = Y(K)+DY2
      IF (Y1-Y(K)) 65,75,70
65    IF (Y1-Y(K+1)) 95,95,75
70    IF (Y1-Y(K+1)) 75,95,95
75    X1 = X(K)+(CO(1,K)+(CO(2,K)+CO(3,K)*DY1)*DY1)*DY1
      IF (X1-XMIN) 80,85,85
80    XMIN = X1
85    IF (X1-XMAX) 95,95,90
90    XMAX = X1
95    IF (Y2-Y(K)) 100,110,105
100   IF (Y2-Y(K+1)) 130,130,110
105   IF (Y2-Y(K+1)) 110,130,130
110   X2 = X(K)+(CO(1,K)+(CO(2,K)+CO(3,K)*DY2)*DY2)*DY2
      IF (X2-XMIN) 115,120,120
115   XMIN = X2
120   IF (X2-XMAX) 130,130,125
125   XMAX = X2
130   IF (XX-XMIN) 530,135,135
135   IF (XX-XMAX) 140,140,530
140   IF (XX-X(K+1)) 150,145,150
145   IF (K+1-KMAX) 530,150,150
150   IF (ABSF(DYN)-ABSF(DXN)) 155,155,160
155   MT = 1
      DX = XX-X(K)
      GO TO 215
160   IF (ABSF(CO(2,K))+ABSF(CO(3,K))) 162,162,164
162   DY=(XX-X(K))/CO(1,K)
      MT=1
      GO TO 210
164   AI(1) = CO(3,K)
      AI(2) = CO(2,K)
      AI(3) = CO(1,K)
      AI(4) = X(K)-XX
      CALL CUBIC (AI,XI,MTYPE)
      IF (MTYPE-1) 166,166,170
166   IF (ABSF(XI(2))-(ABSF(XI(2))-XI(3)**2)) 168,168,170
168   MTYPE=2
170   DX=XX-X(K)
      DO 172 I=1,3
      DY=XI(I)
      DXP=(CO(1,K)+(CO(2,K)+CO(3,K)*DY)*DY)*DY
      D1=CO(1,K)+(2.0*CO(2,K)+3.0*CO(3,K)*DY)*DY
      D2=2.0*CO(2,K)+6.0*CO(3,K)*DY
```

```
172   XI(I)=XI(I)+((DX-DXP)/D1-(DX-DXP)**2/2.0*D2/D1**3)
      MR=0
      DO 178 I=1,MTYPE
      YM=Y(K)+XI(I)
      IF (YM-Y(K)) 174,177,175
174   IF (YM-Y(K+1)) 178,176,177
175   IF (YM-Y(K+1)) 177,176,178
176   IF (K+1-KMAX) 178,177,177
177   MR=MR+1
178   CONTINUE
      MS=MR
      IF (MR) 180,180,182
180   MS=1
182   IF (L+MS-KANS) 184,184,540
184   MT=0
      IF (MR) 186,186,192
186   DO 190 I=1,MTYPE
      DIST=ABSF(XI(I)-(Y(K)+Y(K+1))/2.0)
      IF (I-1) 189,189,188
188   IF (DIST-DMIN) 189,190,190
189   DMIN=DIST
      IMIN=I
190   CONTINUE
      DY=XI(IMIN)
      MT=MT+1
      GO TO 210
192   MS=1
194   YM=Y(K)+XI(MS)
      IF (YM-Y(K)) 195,198,196
195   IF (YM-Y(K+1)) 200,197,198
196   IF (YM-Y(K+1)) 198,197,200
197   IF (K+1-KMAX) 200,198,198
198   DY=XI(MS)
      MT=MT+1
      J=MTYPE-MS
      GO TO 210
200   IF (MS-MTYPE) 205,450,450
205   MS=MS+1
      GO TO 194
210   L = L+1
      CRV(1,L)=XX
      CRV(2,L)=Y(K)+DY
      CRV(3,L)=IXY
      CRV(4,L) = 1.0/(CO(1,K)+(2.0*CO(2,K)+3.0*CO(3,K)*DY)*DY)
      CRV(5,L) = -(2.0*CO(2,K)+6.0*CO(3,K)*DY)*CRV(4,L)**3
      DDXN = 0.5*DY
      GO TO 425
215   L = L+1
      CRV(1,L)=XX
      CRV(2,L)=Y(K)+(CO(1,K)+(CO(2,K)+CO(3,K)*DX)*DX)*DX
      CRV(3,L)=IXY
      CRV(4,L) = CO(1,K)+(2.0*CO(2,K)+3.0*CO(3,K)*DX)*DX
      CRV(5,L) = 2.0*CO(2,K)+6.0*CO(3,K)*DX
      DDXN = 0.5*DX
      GO TO 425
225   ZY = YY
      YMIN = Y(K)
      IF (Y(K+1)-YMIN) 230,235,235
230   YMIN = Y(K+1)
235   YMAX = Y(K)
```

```
      IF (Y(K+1)-YMAX) 245,240,240
240   YMAX = Y(K+1)
245   IF (ABSF(DYN)-ABSF(DXN)) 250,250,330
250   IF (ABSF(CO(2,K))+ABSF(CO(3,K))) 330,330,252
252   IF (CO(3,K)) 254,258,254
254   DISC = CO(2,K)**2-3.0*CO(1,K)*CO(3,K)
      IF (DISC) 330,256,256
256   DX1 = (-CO(2,K)+SQRTF(DISC))/(3.0*CO(3,K))
      DX2 = (-CO(2,K)-SQRTF(DISC))/(3.0*CO(3,K))
      GO TO 260
258   DX1=-0.5*CO(1,K)/CO(2,K)
      DX2 = 0.0
260   X1 = X(K)+DX1
      X2 = X(K)+DX2
      IF (X1-X(K)) 265,275,270
265   IF (X1-X(K+1)) 295,295,275
270   IF (X1-X(K+1)) 275,295,295
275   Y1 = Y(K)+(CO(1,K)+(CO(2,K)+CO(3,K)*DX1)*DX1)*DX1
      IF (Y1-YMIN) 280,285,285
280   YMIN = Y1
285   IF (Y1-YMAX) 295,295,290
290   YMAX = Y1
295   IF (X2-X(K)) 300,310,305
300   IF (X2-X(K+1)) 330,330,310
305   IF (X2-X(K+1)) 310,330,330
310   Y2 = Y(K)+(CO(1,K)+(CO(2,K)+CO(3,K)*DX2)*DX2)*DX2
      IF (Y2-YMIN) 315,320,320
315   YMIN = Y2
320   IF (Y2-YMAX) 330,330,325
325   YMAX=Y2
330   IF (YY-YMIN) 530,335,335
335   IF (YY-YMAX) 340,340,530
340   IF (YY-Y(K+1)) 350,345,350
345   IF (K+1-KMAX) 530,350,350
350   IF (ABSF(DYN)-ABSF(DXN)) 360,360,355
355   MT = 1
      DY = YY-Y(K)
      GO TO 415
360   IF (ABSF(CO(2,K))+ABSF(CO(3,K))) 362,362,364
362   DX=(YY-Y(K))/CO(1,K)
      MT=1
      GO TO 410
364   AI(1) = CO(3,K)
      AI(2) = CO(2,K)
      AI(3) = CO(1,K)
      AI(4) = Y(K)-YY
      CALL CUBIC (AI,XI,MTYPE)
      IF (MTYPE-1) 366,366,370
366   IF (ABSF(XI(2))-(ABSF(XI(2))-XI(3)**2)) 368,368,370
368   MTYPE=2
370   DY=YY-Y(K)
      DO 372 I=1,3
      DX=XI(I)
      DYP=(CO(1,K)+(CO(2,K)+CO(3,K)*DX)*DX)*DX
      D1=CO(1,K)+(2.0*CO(2,K)+3.0*CO(3,K)*DX)*DX
      D2=2.0*CO(2,K)+6.0*CO(3,K)*DX
372   XI(I)=XI(I)+((DY-DYP)/D1-(DY-DYP)**2/2.0*D2/D1**3)
      MR=0
      DO 378 I=1,MTYPE
      XM=X(K)+XI(I)
```

```
      IF (XM-X(K)) 374,377,375
374   IF (XM-X(K+1)) 378,376,377
375   IF (XM-X(K+1)) 377,376,378
376   IF (K+1-KMAX) 378,377,377
377   MR=MR+1
378   CONTINUE
      MS=MR
      IF (MR) 380,380,382
380   MS=1
382   IF (L+MS-KANS) 384,384,540
384   MT=0
      IF (MR) 386,386,392
386   DO 390 I=1,MTYPE
      DIST=ABSF (XI(I)-(X(K)+X(K+1))/2.0)
      IF (I-1) 389,389,388
388   IF (DIST-DMIN) 389,390,390
389   DMIN=DIST
      IMIN=I
390   CONTINUE
      DX=XI(IMIN)
      MT=MT+1
      GO TO 410
392   MS=1
394   XM=X(K)+XI(MS)
      IF (XM-X(K)) 395,398,396
395   IF (XM-X(K+1)) 400,397,398
396   IF (XM-X(K+1)) 398,397,400
397   IF (K+1-KMAX) 400,398,398
398   DX=XI(MS)
      MT=MT+1
      J=MTYPE-MS
      GO TO 410
400   IF (MS-MTYPE) 405,450,450
405   MS=MS+1
      GO TO 394
410   L = L+1
      CRV(1,L)=X(K)+DX
      CRV(2,L)=YY
      CRV(3,L)=IXY
      CRV(4,L)=1.0/(CO(1,K)+(2.0*CO(2,K)+3.0*CO(3,K)*DX)*DX)
      CRV(5,L)=-(2.0*CO(2,K)+6.0*CO(3,K)*DX)*CRV(4,L)**3
      DDXN = 0.5*DX
      GO TO 425
415   L = L+1
      CRV(1,L)=X(K)+(CO(1,K)+(CO(2,K)+CO(3,K)*DY)*DY)*DY
      CRV(2,L)=YY
      CRV(3,L)=IXY
      CRV(4,L) = CO(1,K)+(2.0*CO(2,K)+3.0*CO(3,K)*DY)*DY
      CRV(5,L) = 2.0*CO(2,K)+6.0*CO(3,K)*DY
      DDXN = 0.5*DY
425   DDXN=DDXN/XNBR
      S=0.0
      DO 430 M2=1,INBR
      DXM=2*M2-1
      DXM = DXM*DDXN
430   S = S+SQRTF(1.0+(CO(1,K)+(2.0*CO(2,K)+3.0*CO(3,K)*DXM)*DXM)**2)
      CRV(6,L) = CO(4,K)+2.0*S*ABSF(DDXN)
      IF (J) 450,450,440
440   IF (IXY-1) 200,200,400
450   LO=L-MT+1
```

```
460   IF (MT-1) 500,500,465
465   DO 495 ML=LO,L
      IF (ML-L) 470,500,500
470   SMIN = CRV(6,ML)
      LMIN = ML
      LM = ML+1
      DO 480 M2=LM,L
      IF (CRV(6,M2)-SMIN) 475,480,480
475   SMIN=CRV(6,M2)
      LMIN=M2
480   CONTINUE
      IF (LMIN-ML) 500,500,485
485   XY = CRV(1,LMIN)
      YX = CRV(2,LMIN)
      XYI= CRV(3,LMIN)
      D1 = CRV(4,LMIN)
      D2 = CRV(5,LMIN)
      DO 490 M2=LM,LMIN
      M3=LM+LMIN-M2
      CRV(1,M3)=CRV(1,M3-1)
      CRV(2,M3)=CRV(2,M3-1)
      CRV(3,M3)=CRV(3,M3-1)
      CRV(4,M3)=CRV(4,M3-1)
      CRV(5,M3)=CRV(5,M3-1)
490   CRV(6,M3)=CRV(6,M3-1)
      CRV(1,ML)=XY
      CRV(2,ML)=YX
      CRV(3,ML)=XYI
      CRV(4,ML)=D1
      CRV(5,ML)=D2
      CRV(6,ML)=SMIN
495   CONTINUE
500   CONTINUE
530   IF (KMIN-KMAX) 532,533,533
532   KMIN=KMIN+1
533   K=KMIN
      IF (L-KANS) 535,540,540
535   IF (K-KMAX+1) 20,20,540
540   KFMIN=KMIN
      KFMAX=KMAX
      KFANS=L
550   RETURN
      END (0,1,1,1,0)
```